Separating the User Interface from the Business Logic

Introduction 

This article is the forth in a series of articles I have written for InformIT that describe how to develop script-less, scalable, web applications using Visual Basic 6. In this article, I’ll enhance the sample application to use a 3-tier architecture by splitting the application’s logic into two COM, ActiveX DLL components – one to handle the Web user interface and another to handle the business and data handling logic.

To get the most from this article, you should have already read my previous articles in this series, and be familiar with using ASP, VBScript, and either Microsoft Internet Information Server (IIS) or Microsoft Personal Web Server (PWS) to develop and debug applications for the Web.  I should also add my standard disclaimer that in order to keep the sample application simple to understand, it isn't as robust as a production application should be and I make notes where it is a good idea to include additional logic.  Finally, the code for the sample application is too lengthy to be completely included in this article.  Instead, you can download a zip file containing all the code necessary to run the sample application here
.

3-Tier Architectures

The design goal of 3-tier (a.k.a. multi-tier, n-tier) architectures is to separate an application’s logic into three separate tiers:

· The client tier (a.k.a. user interface tier, user services) is responsible for the presentation of data and interacting with the user.  These days the client tier is likely an HTML-based web application, or a Windows Graphical User Interface (GUI), or both.

· The middle tier (a.k.a. business logic, business services) provides the logic that contains the business rules and also contains the code to interface with the data tier.

· The data-tier (a.k.a. data services) is responsible for data storage and is most often implemented using a relational database management system (RDBMS) such as SQL Server.

The advantages of the 3-tier architecture are widely documented and include:

· Changing the way the data is stored won’t affect the user interface code.

· It can decrease duplicate development effort by allowing the business logic to be shared amongst various types of user interfaces such as a Windows GUI, an HTML-based Web UI, or PDA interface.

· It has the potential to decrease application development calendar time by more easily allowing one developer, or team of developers, to focus on the code for the user interface, and another developer, or team of developers, to focus on the business processing logic for the application.

· It gives you more flexibility with how your servers are configured by allowing you to run code where it is most efficient.  For small businesses, the application components might all run on one server.  For an application that needs to scale, its components can be spread across many servers.  For example, one could be running IIS, a second your Web UI code, a third running the business logic, and a forth running the RDBMS. 

For applications developed with Visual Basic 6, 3-tier architectures are most commonly developed by encapsulating the middle-tier business logic in a COM (a.k.a. ActiveX DLL) component.  This component can be shared with an HTML-based Web UI component running under IIS and ASP as I’ve been doing in this series of articles, and with an executable providing a Windows GUI.

When I first started developing n-tier applications, I searched around for a snappy term and associated acronym for the components that contain the business logic of an application.  The term “business objects” seemed good, but the acronym “BO” didn’t. I found that most of the time these types of components contained at least as much logic for reading and writing data to and from the DBMS as they did business rules, so I started name them “data handling objects” or DHOs for short.  So when you read DHO in this rest of this article, remember I’m talking about a middle-tier, business logic component.

The previous version of the sample application allows the user to search the Publishers table found in the Biblio.mdb database.  In this new version, the application logic will be split into two components.  I’ll keep the existing Web UI component, but move the logic that handles the reading and writing of the Publishers table information to a new PublishersDHO component.  To keep things interesting, I’ll also enhance the application so it allows the user to modify Publisher records.
Component Design Considerations

How do you know which logic goes in which tier?  There is no absolute rule, but a good rule of thumb is if it’s a piece of logic that might be executed no matter what type of UI – web or Windows – then it likely should be coded in the middle tier.

When developing database maintenance applications such as this, I find a good design is to expose two types of objects, or classes, from the DHO for each table involved.  One class encapsulates the table itself.  So as not to confuse this class with the actual database table, or with a VB collection, I call these “list” classes. The other class to expose from the DHO is one that encapsulates a single entity, (i.e. row, or record) from the table. In my sample application, the clsPublishersList class encapsulates the entire Publishers table, and the clsPublisher class encapsulates an individual Publisher entity from the Publishers table. These two classes will work closely together to provide the necessary functionality that allow the applications that use this component to search the Publishers table, and to fetch, insert, update, and delete the data for an individual Publisher.

Component Interface

A server component’s interface is made up of the objects it exposes, along with the properties, methods, and events for those objects.  This interface is all the client code using the component needs to know about, and is the stuff you can see when you open the Object Browser in the Visual Basic IDE and select a referenced component. The sample DHO exposes only two types of objects and we’ve defined a small set of functionality they must provide, so we can now define their interfaces.

	clsPublisherList
	Properties
	None

	
	Methods 
	Public Sub Delete(ByVal lPublisher As clsPublisher)

Removes a Publisher from the Publishers table.

	
	
	Public Sub Insert(ByVal lPublisher As clsPublisher)

Inserts a Publisher into the Publishers table.

	
	
	Public Function Fetch(ByVal lPubID As Long) As clsPublisher

Retrieves a Publisher from the Publishers table.

	
	
	Public Sub Update(ByVal lPublisher As clsPublisher)

Updates a Publisher in the Publishers table.

	
	
	Public Function Search(ByVal lSearchString As String) As Collection 

Returns a new collection containing a list of references to Publishers objects with a Name that matches the search criteria.

	
	Events
	None

	clsPublisher
	Properties
	Public PubID        As Long

Public Name         As String

Public CompanyName  As String

Public Address      As String

Public City         As String

Public State        As String

Public Zip          As String

Public Telephone    As String

Public Fax          As String

The data items stored in the Publishers table.

	
	Methods
	None

	
	Events
	None


Communicating Exceptions

In the previous section I wrote that a component’s interface is made up of the classes it exposes, along with the properties, methods, and events for those classes.  But there is another code communication mechanism that must be used, and this one is a little less obvious to developers who are just beginning to develop components.  

In a typical Windows program, when anything abnormal occurs during processing, it is natural to display a message box to the user to notify him or her about it.  However, when you create a COM component that might be used either from a Windows or Web UI, you must be sure to never code a call to the MsgBox function.  If you do, the message box will be displayed on the computer where the component is installed and running.  In the case of a Web application, this obvious means the message box will be displayed on a server where no one will see it, and not displayed from the user’s web browser client.

So instead of calling the MsgBox function from a component when an exception occurs, the proper thing to do is to use the Visual Basic Err object’s Raise method and hope that the code that the client code using the component has an appropriate error handler defined.  Basically when it comes to handling exceptions, COM components should have the “something unexpected has happened and I can’t continue processing, so whoever called me needs to do something about it” attitude.

The simple syntax of the Err object’s Raise method is:

Call Err.Raise(number, source, description)

where number is a unique identifier for this exception, source is the name of the object that generated the exception, and description is a user-friendly string that describes what has happened.  

In the sample DHO, there is only one case where we really need to raise an error like this.  The component’s interface design includes the clsPublisherList.Fetch method to retrieve the Publisher from the database for the given PubID.  If a record with a matching PubID isn’t in the database, the component will have to raise an error that indicates the PubID was not found.  The syntax of Err.Raise requires an error number to be supplied, so what value should be used?  To be guarantee the number you choose for the error number isn’t the same as an error number already used by something else, Visual Basic includes the vbObjectError constant which you use to add to your own internal error number.  For example, to raise an error indicating a Publisher with the given PubID wasn’t found in the database, you might code:

Call Err.Raise(vbObjectError + 1, "clsPublisher", _

    "PubID " & CStr(PubID) & " not found.")

But to give the code that uses your component an easy method of determining what kind of exception occurred, we should associate constants with the actual error number values and make these constants available from outside of the component.  These constants then also become part of the component’s interface.  I find enumerated types work very nice for this. 

I suggest that all errors a given class might raise be defined as Public Enums in the class module’s General Declaration section.  For example, if clsPublisher could raise a not found error and a duplicate key error, we might define constants for these errors like this:

Public Enum PublisherErrors

    peNotFound = vbObjectError + 1

    peDupPubID

End Enum

And then raise an error like this:

Call Err.Raise(peNotFound, "clsPublisher", _

    "PubID " & CStr(PubID) & " not found.")

Unfortunately, even if you define the error numbers as Public Enums in a class module, they don’t really belong to that class module and are instead treated as global scope.  So if your code calls a method in a component and want’s to test for a specific error, you write this:


If Err.Number = peNotFound Then …

or prefixed it with the component name like this:


If Err.Number = PublishersDHO.peNotFound Then …

but you don’t prefix it with the class name like this:


If Err.Number = clsPublisher.peNotFound Then …

That’s why you should use some sort of prefixing scheme to attempt to make your error number constants unique to the class and/or component they are raised from.

Create the DHO COM Project

Use Visual Basic to open the previous PublishersWebUI project, and choose Add Project from the File menu.  From the Add Project dialog, choose AxtiveX DLL to create a project file for the new DHO component.  Visual Basic will create a new project with a default class module (Class1) and a new project group.  Delete this newly created class from the project by selecting it in the Project Explorer window and choosing Remove from the Project menu.  

If you haven’t worked with project groups before, here’s a short explanation.  You know that a VBP project file contains a list of forms, classes, and code modules that make up a project.  Similarly a project group file contains a list of projects.  Project group files have a VBG file type and are really handy when developing a multi-component application.  Just double-click the VBG file and the Visual Basic IDE automatically opens up everything you need.

Select the new project in the Project Explorer window, and choose Properties from the Project window.  At a minimum, use the Project Properties dialog to set the Project Name, Project Description, and Application Title to something appropriate such as “PublishersDHO”.  Close the dialog and save the project file by selecting Save PublishersDHO from the File menu.  I suggest saving the DHO project files in a separate folder from the Web UI project files.  Finally, save the project group by selecting Save Project Group As from the file menu, and give it a file name like “PublishersWebUI.VBG”.

Recall that the purpose of the DHO component is to do the actual reading and writing of the data to the database so the PublishersWebUI project doesn’t have to.  Database connectivity is handled through ADO so you have to add a reference to the Microsoft ActiveX Data Object Library to the PublishersDHO project.  You must also modify the PublishersWebUI project to remove the reference to the Microsoft ActiveX Data Object Library, and add a reference to the new PublishersDHO component.  Do this by selecting References from the Project menu.

Code the Class Modules

From the Project menu, choose Add Class Module and select Class Module from the Add Class Module dialog, change the Name property of this new class to clsPublishersList, and save as PublishersList.cls.  Repeat this procedure to add a class for clsPublishers.  Make sure the Instancing property for both classes is set to the default value of 5-MultiUse.  MultiUse allows multiple objects of this class to be instantiated outside of the component.  You’re now ready to start writing the code to implement the necessary features.  

My design has clsPublisherList and clsPublisher working closely together to provide the functionality that allows the client code that use this component to search the Publishers table, and fetch, insert, update, and delete individual Publishers.  Exactly how this is done is not quite as important as providing a correct component interface, because the encapsulation rules of object-oriented programming say its alright to change the way something is implemented as long as you don’t change the interface. I usually put the code that works with more than one entity (e.g. the Search method) in the list class (e.g. clsPublisherList). The methods that work with a single entity (e.g. Fetch, Insert, Update, and Delete) are then normally delegated to the class that encapsulates the notion of one record in the table (e.g. clsPublisher).

I also make the List class responsible for using ADO to open and close the database connection. clsPublishersList can do this automatically if you add code to the Class_Intialize and Class_Terminate events:

' Private PublisherList variables.

Private mDB As ADODB.Connection

' Automatically open a database connection when a PublisherList

' object is created.

Private Sub Class_Initialize()

    Dim lConnectString As String

    lConnectString = "Driver={Microsoft Access Driver (*.mdb)};" _

        & "DBQ=" & App.Path & "\Biblio.mdb"

    Set mDB = New ADODB.Connection

    Call mDB.Open(lConnectString)

End Sub

' Automatically close the database connection when a PublisherList

' object is destroyed.

Private Sub Class_Terminate()

    Call mDB.Close

    Set mDB = Nothing

End Sub

Another method implemented by clsPublisherList is the Search method.  Its job is to query the database for all Publishers whose name matches the search criteria passed as an argument.  I code it like this:

' Returns a new collection containing a list of references to

' Publishers objects with a Name that matches the search criteria.

Public Function Search(ByVal lSearchString As String) As Collection

    Dim lSQL          As String

    Dim lPublisher    As clsPublisher

    Dim lPublisherCol As Collection

    Dim lRS           As ADODB.Recordset

    ' Build the SQL statement.

    lSQL = "SELECT * FROM Publishers " _

        & " WHERE Name LIKE " & DBStr("%" & lSearchString & "%") _

        & " ORDER BY Name"

    ' Query the database for the matching Publishers and create a

    ' new clsPublisher object for each record returned.

    Set lPublisherCol = New Collection

    Set lRS = mDB.Execute(lSQL)

    Do Until lRS.EOF = True

        Set lPublisher = New clsPublisher

        Call lPublisher.RSGet(lRS)

        Call lPublisherCol.Add(lPublisher, CStr(lPublisher.PubID))

        Call lRS.MoveNext

    Loop

    Call lRS.Close

    Set Search = lPublisherCol

End Function

Whenever you have SQL statements embedded in your source code that contains strings entered by the user, you always have the potential the user will enter an apostrophe (', a.k.a. single-quote) somewhere in the string.  Because this same apostrophe character is used as a delimiter within the SQL statement, it will generate SQL syntax errors if you don’t write code to do something about it.  I call this the “SQL embedded quote problem”.  The ugly solution to this problem is to modify the user interfaces so they do not allow apostrophes to be entered in string fields. However, in my opinion, a much better solution is to search through the strings when the SQL statements are built and replace all apostrophes with two apostrophes. The DBMS then interprets the two apostrophes as just one.  Since this must be done for every string passed to the database, I wrap this code in a function named DBStr and code it in a standard code module so it can be used from anywhere inside the component.  For example:

' Replace each apostrophe with two apostrophes.

Public Function DBStr(ByVal lStr As String) As String

    DBStr = "'" & Replace(lStr, "'", "''") & "'"

End Function

As coded above, the Search method also relies on the clsPublisher.RSGet method to retrieve its property values from the ADO Recordset object.  I put this chunk of code in the clsPublisher module because it is working with only one Publisher entity:

Friend Sub RSGet(ByVal lRS As ADODB.Recordset)

    ' Note that concatenating an empty string to the

    ' recordset fields prevents any errors caused by

    ' null fields.

    PubID = lRS!PubID

    Name = "" & lRS!Name

    CompanyName = "" & lRS![Company Name]

    Address = "" & lRS!Address

    City = "" & lRS!City

    State = "" & lRS!State

    Zip = "" & lRS!Zip

    Telephone = "" & lRS!Telephone

    Fax = "" & lRS!Fax

End Sub

Note that RSGet is declared using the Friend keyword.  A procedure declared Friend can be called from anywhere within the component, but is not visible outside of the component.  In this sample code, it means that RSGet can be called from clsPublisher and clsPublisherList, but it can’t be called from anywhere in the PublishersWebUI project.  This is good because we don’t want any code outside of the DHO to query the Publishers table.  Instead, we want it to use the DHO’s methods.

The methods in clsPublishersList that allow individual Publishers to be fetched, inserted, updated, and deleted are pretty simple to code because they just pass control to “friend” methods in clsPublisher:

' Removes a Publisher from the Publishers table.

Public Sub Delete(ByVal lPublisher As clsPublisher)

    Call lPublisher.DBDelete(mDB)

End Sub

' Inserts a Publisher into the Publishers table.

Public Sub Insert(ByVal lPublisher As clsPublisher)

    Call lPublisher.DBInsert(mDB)

    ' If we were worried about duplicate primary keys,

    ' we would attempt to catch the error raised by

    ' the DBMS here, and re-raise a nicer message.

End Sub

' Retrieves a Publisher from the Publishers table.

Public Function Fetch(ByVal lPubID As Long) As clsPublisher

    Dim lPub As clsPublisher

    Set lPub = New clsPublisher

    lPub.PubID = lPubID

    Call lPub.DBGet(mDB)

    Set Fetch = lPub

End Function

' Updates a Publisher in the Publishers table.

Public Sub Update(ByVal lPublisher As clsPublisher)

    Call lPublisher.DBUpdate(mDB)

End Sub

Because we’ll add Publisher update functionality in the sample application, here’s the code for the clsPublisher.DBUpdate method:

Friend Sub DBUpdate(ByVal lDB As ADODB.Connection)

    Dim lSQL As String

    lSQL = "UPDATE Publishers SET "

    lSQL = lSQL & " Name = " & DBStr(Name) & ","

    lSQL = lSQL & " [Company Name] = " & DBStr(CompanyName) & ","

    lSQL = lSQL & " Address = " & DBStr(Address) & ","

    lSQL = lSQL & " City = " & DBStr(City) & ","

    lSQL = lSQL & " State = " & DBStr(State) & ","

    lSQL = lSQL & " Zip = " & DBStr(Zip) & ","

    lSQL = lSQL & " Telephone = " & DBStr(Telephone) & ","

    lSQL = lSQL & " Fax = " & DBStr(Fax)

    lSQL = lSQL & " WHERE PubID = " & CStr(PubID)

    Call lDB.Execute(lSQL)

End Sub

Note that this procedure also is coded using the Friend keyword so that all request to update a publisher must go through the Public clsPublisherList.Update method first.  clsPublisher.DBUpdate also uses the DBStr function mentioned earlier to avoid the SQL embedded quotes problem.

The DBFetch, DBDelete, and DBInsert methods of clsPublisher are coded very much the same as DBUpdate and so aren’t listed here for brevity sake.  Download the sample project source code if you want to see how they are implemented.

Change the Web UI Component to use the DHO Component

By moving all the business and data handling logic to the DHO, the Web UI code becomes much simpler – no more ADO database and recordset objects and SQL statements to mess with!  For example, the Web UI component contains a procedure named DetailProc which needs to get the Publisher indicated by the PubID value on the querystring.  To get a single Publisher using the new DHO, DetailProc now only needs four lines of code. 

Dim lPubList As clsPublisherList

Dim lPub     As clsPublisher 

Set lPubList = New clsPublisherList

Set lPub = lPubList.Fetch(gAspReq.QueryString.Item("PubID"))

And the code in the SearchProc procedure can get a Collection of all Publishers that match the search string by calling the clsPublisherList.Search method:

Dim lPubList As clsPublisherList

Dim lPubCol  As Collection

' Get the Filter value from the form.

mUserState.Filter = gAspReq.Item("txtFilter")

' Instantiate the clsPublisherList object and ask it for a list

' of Publishers with a name that matches the search filter.

Set lPubList = New clsPublisherList

Set lPubCol = lPubList.Search(mUserState.Filter)

Of course there are now more total lines of code that get executed if you count the code in both the PublishersWebUI and PublishersDHO components, but the hope is that the advantages listed at the beginning of this article outweigh any performance hit this technique creates.

I’ve also enhanced the sample application by changing the detail form to allow Publisher data to be updated.  This basically involves change the Detail.htm file to use a form, and adding a new UpdateProc action to the Select Case statement in the clsWebApp.WebMain method.  Again for brevity sake, the entire UpdateProc procedure is not listed in this article, but the code that uses the DHO component to do the update is now pretty trivial:

' Create a new Publisher object.

Set lPub = New clsPublisher

' Get the new values from the form.  Example:

lPub.PubID = gAspReq.QueryString("PubID")

lPub.Name = gAspReq.Form("txtName")

…

' Ask the clsPublisherList object to update this Publisher.

Set lPubList = New clsPublisherList

Call lPubList.Update(lPub)

Summary

Developing an application using a 3-tier architecture design requires some additional up-front design and coding time, but this time is well worth it in the long run because of the advantages you gain from separating the UI code from the business and data handling logic.  In this article, I’ve demonstrated one design I’ve found effective for accomplishing this.  In the next article in this series, I’ll enhance the Web UI to allow new Publishers to be added and existing ones to be deleted, and show how you can protect your data from errors that might be caused by multi-user concurrency problems.

